
* Stata Analysis Examples Replication for ASDA 2nd Edition   
* Berglund April 2017  
* Chapter 10  
 
use "P:\ASDA 2\Data sets\ncsr\ncsr_sub_5apr2017.dta", clear 
rename *, lower 
 
gen ageonsetmde=age  
replace ageonsetmde=mde_ond if mde==1  
tab ageonsetmde  
gen intwage=age  
 
* set survey time  
stset ageonsetmde [pweight = ncsrwtsh], failure(mde==1) 
 
* Section 10.3.3 KM Example  
sts graph, by(racecat) legend(lab(1 "Other/Asian") lab(2 "Hisp/Mexican") lab(3 "Black") lab(4 
"White")) 
sts list, at(10 20 30 40 50 60 70) 
 
* 10.4.5 Example: Fitting a Cox Proportional Hazards Model to Complex Sample Survey Data 
gen swd=0  
replace swd=1 if mar3cat==2  
gen nevermarried=0  
replace nevermarried=1 if mar3cat==3  
 
gen ed12=0  
replace ed12=1 if ed4cat==2  
gen ed1315=0  
replace ed1315=1 if ed4cat==3  
gen ed16=0  
replace ed16=1 if ed4cat==4  
 
* set survey variables and time  
svyset seclustr [pweight=ncsrwtsh], strata(sestrat) vce(linearized) singleunit(missing)  
stset ageonsetmde [pweight = ncsrwtsh], failure(mde==1) 
* Table 10.2 
svy: stcox intwage ib2.sex i.mar3cat i.ed4cat i.racecat 
 
* Figure 10.4  
stphplot, by(racecat) legend(lab(1 "Other/Asian") lab(2 "Hisp/Mexican") lab(3 "Black") lab(4 
"White"))  
 
* Figure 10.5  
stphplot, by(racecat) adj(intwage sexf swd nevermarried ed12 ed1315 ed16) nonegative nolntime /// 
legend(lab(1 "Other/Asian") lab(2 "Hisp/Mexican") lab(3 "Black") lab(4 "White"))  
 
*10.5.5 Fitting a Discrete Time Model to Complex Sample Survey Data 
* prepare data for model, shape into "long" file   
expand intwage  
bysort caseid: gen pyr = _n  
 
gen mdetv = 1 if pyr == mde_ond  
replace mdetv = 0 if pyr != mde_ond   
tab mdetv  
 
list caseid intwage ncsrwtsh sestrat seclustr pyr mdetv ageonsetmde if caseid == 1  
* set survey variables and run logistic model for Table 10.5  
svyset seclustr [pweight=ncsrwtsh], strata(sestrat) vce(linearized) singleunit(missing)   
svy: logistic mdetv pyr intwage ib2.sex i.ed4cat i.racecat i.mar3cat if pyr <= ageonsetmde 
* Table 10.6 CLOGLOG model  
svy: cloglog mdetv pyr intwage ib2.sex i.ed4cat i.racecat i.mar3cat if pyr <= ageonsetmde, eform  


